
Selected SAS Functions Reference
ABS  –  Numeric Function
Author:  Steve Jost
Purpose: Return the absolute value of the input parameter.
Input Parameter:  numeric arg. 
Return Value:  numeric.  
SAS Test Script:
data abs;
   input a @@;
   b = abs(a);
   datalines;
45 -23  0  2343.3  -3343.2  8.9e-32  -4.3e+15
;
proc print;
Output from Test Script:

                  Obs            a           b

                   1          45.0        45.0
                   2         -23.0        23.0
                   3           0.0         0.0
                   4        2343.3      2343.3
                   5       -3343.2      3343.2
                   6           0.0         0.0
                   7       -4.3E15      4.3E15

LENGTH, LENGTHN, LENGTHC – String Functions

Author: Steve Jost

Purpose: Return the number of characters in a character string (more or less).  
   LENGTH returns the length of the input string minus any trailing blanks, except when the 
      string is a missing value (zero length string).  In that case, LENGTH returns 1.  
   LENGTHN also returns the length of the input string minus any trailing blanks.  However, it returns 0 for a string that is all blanks.string of all blanks.
   
   LENGTHC returns the total storage space for a string, including trailing blanks.  It returns 1 for a zero length string. 

Input Parameter: string arg.

Return Value: numeric.

SAS Test Script:

data abs;
   input @1 a $8. @1 b $CHAR.;
   lengtha = length(a);
   lengthb = length(b);
   lengthna = lengthn(a);
   lengthnb = lengthn(b);
   lengthca = lengthc(a);
   lengthcb = lengthc(b);
   datalines;
elephant 
horse 
dog

;
proc print;
Test Script Output:

                                           l    l    l    l
                                 l    l    e    e    e    e
                                 e    e    n    n    n    n
                                 n    n    g    g    g    g
                                 g    g    t    t    t    t
    O                            t    t    h    h    h    h
    b                            h    h    n    n    c    c
    s    a           b           a    b    a    b    a    b

    1    elephant    elephant    8    8    8    8    8    8
    2    horse       horse       5    5    5    5    8    8
    3    dog         dog         3    3    3    3    8    8
    4                            1    1    0    0    8    8



MEAN  VAR  STD  STDERR – Numeric Functions
Author: Steve Jost
Purpose: Return the sample mean, sample variance, sample standard deviation, or standard error, respectively, of the input list.
Input parameters:  list of numeric values.  
Return value:  numeric.  
SAS Test Script:
data orig;
   input x @@;
   datalines;
34  65  .  33   8  19  89  20  .
;
proc means n mean var std stderr;

data stats;
   m = mean(34, 65, ., 33, 8, 19, 89, 20,  .);
   v = var(34, 65, ., 33, 8, 19, 89, 20,  .);
   sd = std(34, 65, ., 33, 8, 19, 89, 20,  .);
   se = stderr(34, 65, ., 33, 8, 19, 89, 20,  .);
   output;
proc print;
Output from Test Script:

                      The MEANS Procedure

                     Analysis Variable : x

 N           Mean       Variance        Std Dev      Std Error
 ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
 7     38.2857143    825.9047619     28.7385588     10.8621542
 ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        Obs       m          v          sd         se

         1     38.2857    825.905    28.7386    10.8622

SUBSTR – String Function
Author: Steve Jost
Purpose: Return a substring of the input string.
Input parameters:   1. arg: string, position: numeric, length: numeric
                                    2. arg: string, position: numeric
Return value:  string.   Returns the substring of the input string of the specified length that starts at the character specified by position.  For example, if the input string is "elephant", then substr("elephant", 3, 4) returns "epha".  If the length is not specified, then the returned substring continues to the end of the string.  For example substr("elephant", 3) returns ephant.  Also, if the length would produce a substring that continues past the end of the string, no error is produced.  The resulting string is the same as the one produced without the length parameter. 
Note that index for the position starts at character 1 at the left  (not at 0 as most other languages like Java, Python, C++, C#, and Ruby). 
SAS Test Script:  
data orig;
[bookmark: _GoBack]   input str $ pos len @@;
   if len = . then sub = substr(str, pos);
   else sub = substr(str, pos, len);
   datalines;
elephant 3 4
elephant 3 15
elephant 3 .
;
proc print;
Output from Test Script:

            Obs      str       pos    len     sub

             1     elephant     3     4      epha
             2     elephant     3     15     ephant
             3     elephant     3            ephant



SCAN – String Function
Author: Steve Jost
Purpose:  Scan an input string to words.  For the purposes of the SCAN function, a word is a sequence of characters that is bounded on the left by a delimiter or the beginning of the input string, (2) bounded on the right by a delimiter or the end of the string, (3) itself contains no delimiters, unless the Q modifier applies (see below).  Finding words within an input string is harder than finding substrings with the SUBSTR function, which defines a substring by its start position and length.
Input Parameters:  
   arg: string – The input string to be scanned to find a substring;
   count: numeric -- The number of the delimited item to be found.  A positive number means 
      scan the input string from left to right.  A negative value means scan the input string from  
      right to left.
   charlist: string, initializes a list of delimiters.  Some of the following options can add to the list
      of delimiters or otherwise modify the behavior of the scan function.
   modifier:  characters that modify the behavior of the delimiters.  Here is the list of possible 
      modifiers:
         A or a: Adds alphabetic characters to the delimiters in charlist.
         B or b: Scans from right to left, regardless of the sign of count.
         C or c: Adds control characters to the delimiters in charlist.
         D or d: Adds digits to the delimiters in charlist.
         F or f: Adds an underscore and English letters (valid characters for the beginning of a SAS 
            variable name) to the delimiters in charlist.
         H or h: Adds a horizontal tab to the delimiters in charlist.
         I or i: Ignore the case of letters.
         K or k: Causes all characters not in charlist to be treated as delimiters.
         L or l: Adds lower case letters to the delimiters in charlist.
         M or m: Specifies that multiple delimiters or delimiters at the beginning of the input string 
            arg to be treated as zero length strings.  Otherwise multiple delimiters or delimiters at 
            the beginning or end of the input string are ignored.
         N or n: Adds an underscore, digits, and English letters (valid characters for a SAS variable 
            name) to the delimiters in charlist.
         O or o: causes the charlist and modifier arguments to be processed only once, rather than 
            every time the SCAN function is called.  This can reduce the execution time in loops.
         P or p: Adds punctuation marks ( . , : ; ! ? ) to the list of delimiters in charlist.
         Q or q: Ignores delimiters that are inside of quotation marks.
         R or r: Removes leading and trailing blanks from the input string arg to be scanned.
         S or s: Adds white space characters (blank, horizontal tab, vertical tab, carriage return, line 
            feed, form feed) to the list of delimiters in charlist.
         T or t: Trims trailing blanks from the input string arg to be scanned.
         U or u: Adds upper case letters to the delimiters in charlist.
         W or w: Adds writable (printable) characters to the delimiters in charlist.
         X or x: Adds hexadecimal characters (0 1 2 3 4 5 6 7 8 9 A B C D E F a b c d e f) to the 
            delimiters in charlist.
         
SAS Test script
* Note: the modifier options are 
* not tested in this test script;
data testscan;
   input arg $ 1-45;
   if _n_ = 1 then delim = ' ';
   else if _n_ = 2 then delim = ';';
   do i = 1 to 9;
      word = scan(arg, i, delim);
      len = lengthn(word);
      output;
   end;
   keep word delim len;
   datalines4;
The quick brown fox jumped over the lazy dog.
cat;horse;wolf;giraffe;elephant
;;;;
proc print;
TestScript Output

                Obs    delim    word        len

                  1             The          3
                  2             quick        5
                  3             brown        5
                  4             fox          3
                  5             jumped       6
                  6             over         4
                  7             the          3
                  8             lazy         4
                  9             dog.         4
                 10      ;      cat          3
                 11      ;      horse        5
                 12      ;      wolf         4
                 13      ;      giraffe      7
                 14      ;      elephant     8
                 15      ;                   0
                 16      ;                   0
                 17      ;                   0
                 18      ;                   0 

PCTL1  PCTL2  PCTL3  PCTL4  PCTL5 – Numeric Function
Author: Steve Jost
Purpose: Compute the specified percentile of an input list.  SAS has 5 different algorithms that are available for computing percentiles, specified by PCTLn, n = 1, 2, 3, 4, 5.
Input Parameters: 
   percentile: numeric -- requested percentile. 
   input list: numeric – list of data for which percentile is to be computed.
Return Value: numeric.

SAS Test Script:
* Change the pctldef to 2, 3, 4, 5
  to get other definitions of pecentiles;
data orig;
   input x @@;
   datalines;
45.5 76.9 22.2  3.4 84.4 98.9
;
proc means p25 p50 p75 pctldef=1 maxdec=3;

data percentiles;
   p25 = pctl1(25, 45.5, 76.9, 22.2, 3.4, 84.6, 98.9);
   p50 = pctl1(50, 45.5, 76.9, 22.2, 3.4, 84.6, 98.9);
   p75 = pctl1(75, 45.5, 76.9, 22.2, 3.4, 84.6, 98.9);
proc print;



Test Script Output:

                      The MEANS Procedure

                     Analysis Variable : x

             25th Pctl       50th Pctl       75th Pctl
          ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
                12.800          45.500          80.750
          Ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                  Obs     p25     p50     p75

                   1     12.8    45.5    80.75

